```
> restart:
> with(OreModules):
> with(OreMorphisms);
> with(linalg):
```

Let us consider the Cauchy-Riemann equations defined by the following matrix of differential operators in d_x and d_y (see, e.g., R. Courant, D. Hilbert, *Methods of Mathematical Physics*, Wiley Classics Library, Wiley, 1989):

```
> A:=DefineOreAlgebra(diff=[d[x],x],diff=[d[y],y],polynom=[x,y]):
```

```
> R:=matrix(2,2,[d[x],-d[y],d[y],d[x]]);
```

$$R := \begin{bmatrix} d_x & -d_y \\ d_y & d_x \end{bmatrix}$$

Let us introduce the $A = \mathbb{Q}[d_x, d_y]$-module $M = A^{1\times 2}/(A^{1\times 2}\, R)$ finitely presented by the matrix R. The endomorphism ring $E = \mathrm{end}_A(M)$ of M is then defined by:

```
> Endo:=MorphismsConstCoeff(R,R,A,mult_table);
```

$$Endo := [[\begin{bmatrix} 0 & 1 \\ -1 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}], \begin{bmatrix} -d_y & d_x \\ d_x & d_y \end{bmatrix}, \begin{bmatrix} [1,1] & 0 & -1 \\ [1,2] & 1 & 0 \\ [2,1] & 1 & 0 \\ [2,2] & 0 & 1 \end{bmatrix}]$$

The A-module E is finitely generated by the endomorphisms f_i's defined by $f_i(\pi(\lambda)) = \pi(\lambda\, P_i)$, where $\pi : A^{1\times 2} \longrightarrow M$ denotes the projection onto M, $\lambda \in A^{1\times 2}$ and P_i is one of the two matrices defined in $Endo[1]$, i.e., $f_1(\pi(\lambda_1 \quad \lambda_2)) = \pi((-\lambda_2 \quad \lambda_1))$ and $f_2 = \mathrm{id}_M$. If we now denote by $F = (f_1 \quad f_2)^T$, then the generators f_i's of E satisfy the relations $Endo[2]\, F^T = 0$, where $Endo[2]$ denotes the second entry of $Endo$. The previous data completely characterize the A-module structure of E and shows that $E \cong M$. The ring structure of the endomorphism ring E is defined by the multiplication table of the generators f_i's of E, namely, the knowledge of the expressions of the products $f_i \circ f_j$ on the family of generators $\{f_1, \mathrm{id}_M\}$ of E. More precisely, if we denote by $\otimes$ the Kronecker product, namely, $F \otimes F = ((f_1 \circ F)^T \quad (\mathrm{id}_M \circ F)^T)^T$, then the multiplication table T is defined by $F \otimes F = T\, F$, where T denotes the matrix $Endo[3]$ without the first column which corresponds to the indices (i, j) of the product $f_i \circ f_j$. We obtain:

$$f_1 \circ f_1 = -\mathrm{id}_M, \quad f_1 \circ \mathrm{id}_M = \mathrm{id}_M \circ f_1 = f_1, \quad \mathrm{id}_M \circ \mathrm{id}_M = \mathrm{id}_M. \tag{1}$$

We now study the idempotents of the ring E. As every endomorphism f of E has the form $f = a_1\, f_1 + a_2\, \mathrm{id}_M$, where a_1 and $a_2 \in A$, using the multiplication table (1), we obtain that $f^2 = f$ is equivalent to $(2\, a_1\, a_2 - a_1)\, f_1 + (a_2^2 - a_1^2 - a_2)\, \mathrm{id}_M = 0$. If we only consider idempotent endomorphisms defined by constant matrices (i.e., matrices formed by zero-order differential operators), i.e., a_1 and a_2 are constants, using the fact that f_1 and id_M do not satisfy zero-order differential relation, then we have

$$\begin{cases} -a_1^2 + a_2^2 - a_2 = 0, \\ 2\, a_1\, a_2 - a_1 = 0, \end{cases} \Leftrightarrow (a_1, a_2) \in \{(0,0), (0,1), (i/2, 1/2), (-i/2, 1/2)\},$$

i.e., $e_1 = 0$, $e_2 = \mathrm{id}_M$, $e_3 = (i\, f_1 + \mathrm{id}_M)/2$ and $e_4 = (-i\, f_1 + \mathrm{id}_M)/2$. Let us check this result:

```
> Idem:=IdempotentsConstCoeff(R,Endo[1],A,0,alpha);
```

$$Idem := [[\begin{bmatrix} 1/2 & 1/2\,\alpha_1 \\ -1/2\,\alpha_1 & 1/2 \end{bmatrix}, \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}], [Ore\_algebra, [\text{“}diff\text{''}, \text{“}diff\text{''}], [x,y],$$
$$[d_x, d_y], [x,y], [\alpha_1], 0, [], [{\alpha_1}^2 + 1], [x,y], [], [], [diff = [d_x, x], diff = [d_y, y]]]]$$

We obtain that there only exist the two trivial idempotents 0 and id_M of E defined by constant matrices over A but one non-trivial idempotent e defined over the ring $B = \mathbb{Q}[\alpha_1]/(x_1^2+1)[d_x, d_y]$ (i.e., $B = \mathbb{Q}(i)[d_x, d_y]$), i.e., $e \in F = \mathrm{end}_B(B \otimes_A M)$ is defined by

$$\forall\, \lambda \in B^{1\times 2}, \quad e((\mathrm{id}_B \otimes \pi)(\lambda)) = (\mathrm{id}_B \otimes \pi)(\lambda\, P),$$

where the matrix $P \in B^{2\times 2}$ is given by:

```
> B:=Idem[2]: P:=Idem[1,1]; Q:=Factorize(Mult(R,P,B),R,B);
```

$$P := \begin{bmatrix} 1/2 & 1/2\,\alpha_1 \\ -1/2\,\alpha_1 & 1/2 \end{bmatrix} \quad Q := \begin{bmatrix} 1/2 & 1/2\,\alpha_1 \\ -1/2\,\alpha_1 & 1/2 \end{bmatrix}$$

The matrix $Q \in B^{2\times 2}$ satisfying $R\,P = Q\,R$ is equal to P. We can check that $P^2 = P$:

```
> VERIF_IDEMPOTENT:=subs(alpha[1]^2=-1,simplify(evalm(Mult(P,P,B)-P)));
```

$$\mathit{VERIF\_IDEMPOTENT} := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

As the entries of P belong to the field $\mathbb{Q}(i) = \mathbb{Q}[\alpha_1]/(\alpha_1^2+1)$, using linear algebraic techniques, we can easily find bases of the free $\mathbb{Q}(i)$-modules $\ker_{\mathbb{Q}(i)}(.P)$ and $\mathrm{im}_{\mathbb{Q}(i)}(.P) = \ker_{\mathbb{Q}(i)}(.(I_2 - P))$, and thus, bases of the free B-modules $\ker_B(.P)$ and $\mathrm{im}_B(.P)$:

```
> U1:=SyzygyModule(P,B): U2:=SyzygyModule(evalm(1-P),B):
> U:=stackmatrix(U1,U2);
```

$$U := \begin{bmatrix} \alpha_1 & 1 \\ \alpha_1 & -1 \end{bmatrix}$$

We can check that the matrix $U\,P\,U^{-1}$ is the block-diagonal matrix $\mathrm{diag}(0,1)$:

```
> VERIF:=subs(alpha[1]^2=-1,Mult(U,P,LeftInverse(U,B),B));
```

$$\mathit{VERIF} := \begin{bmatrix} 0 & 0 \\ 0 & 1 \end{bmatrix}$$

Then, we know that the matrix is equivalent to the block-diagonal matrix $U\,R\,U^{-1}$ defined by:

```
> R_dec:=subs(alpha[1]^2=-1,map(collect,Mult(U,R,LeftInverse(U,B),B),d[y]));
```

$$\mathit{R\_dec} := \begin{bmatrix} \frac{d_y}{\alpha_1} + d_x & 0 \\ 0 & -\frac{d_y}{\alpha_1} + d_x \end{bmatrix}$$

The last result can be directly obtained by means of the command *HeuristicDecomposition*:

```
> S:=subs(alpha[1]^2=-1,map(collect,HeuristicDecomposition(R,P,B)[1],d[y]));
```

$$S := \begin{bmatrix} \frac{d_y}{\alpha_1} + d_x & 0 \\ 0 & -\frac{d_y}{\alpha_1} + d_x \end{bmatrix}$$

If we substitute $\alpha_1 = i$ into the previous block-diagonal matrix, we then obtain

```
> subs(alpha[1]=I,evalm(S));
```

$$\begin{bmatrix} -i\,d_y + d_x & 0 \\ 0 & i\,d_y + d_x \end{bmatrix}$$

i.e., $S = \mathrm{diag}(\overline{\partial}, \partial)$, with the standard notations $\partial = d_x + i\,d_y$ and $\overline{\partial} = d_x - i\,d_y$.

Similarly, we can consider the following matrix of differential operators in d_t and d_x

```
> A:=DefineOreAlgebra(diff=[d[t],t],diff=[d[x],x],polynom=[t,x],comm=[a,b]):
```

```
> R:=matrix(2,2,[d[x],a*d[t],d[t],b*d[x]]);
```

$$R := \begin{bmatrix} d_x & a\,d_t \\ d_t & b\,d_x \end{bmatrix}$$

where a and b denote two real parameters. Let us introduce the $A = \mathbb{Q}(a,b)[d_t, d_x]$-module $M = A^{1\times 2}/(A^{1\times 2}\,R)$. If we consider an A-module $\mathcal{F}$ (e.g., $\mathcal{F} = C^{\infty}(\mathbb{R}^2)$), then the linear system of differential equations $\ker_{\mathcal{F}}(R.) \cong \hom_A(M, \mathcal{F})$ corresponds, for instance, to an acoustic wave ($a = 1/\rho$, $b = \rho\, c^2$) or a LC transmission line ($a = L$, $b = 1/C$).

Let us compute the endomorphism ring $E = \mathrm{end}_A(M)$ of M:

```
> Endo:=MorphismsConstCoeff(R,R,A,mult_table);
```

$$Endo := [[\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & a\,b \\ 1 & 0 \end{bmatrix}], \begin{bmatrix} b\,d_x & d_t \\ a\,d_t & d_x \end{bmatrix}, \begin{bmatrix} [1,1] & 1 & 0 \\ [1,2] & 0 & 1 \\ [2,1] & 0 & 1 \\ [2,2] & a\,b & 0 \end{bmatrix}]$$

We now compute idempotents of E defined by constant idempotent matrices:

```
> Idem:=IdempotentsMatConstCoeff(R,Endo[1],A,0,alpha);
```

$$Idem := [\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, [\begin{bmatrix} 1/2 & \alpha_1\,a\,b \\ \alpha_1 & 1/2 \end{bmatrix}], [Ore\_algebra, [\text{``}diff\text{''}, \text{``}diff\text{''}], [t,x],$$
$$[d_t, d_x], [t,x], [a, b, \alpha_1], 0, [\,], [-1 + 4\,{\alpha_1}^2\,a\,b], [t,x], [\,], [\,], [diff = [d_t, t], diff = [d_x, x]]]]$$

We obtain the two trivial idempotents of 0 and id_M of E. However, if we consider the ring $B = \mathbb{Q}(a,b)[\alpha_1]/(4\,a\,b\,\alpha_1^2 - 1)[d_t, d_x]$, then a non-trivial idempotent of $\mathrm{end}_B(B \otimes_A M)$ is defined by the following matrix:

```
> B:=Idem[2]: P:=Idem[1,1];
```

$$P := \begin{bmatrix} 1/2 & \alpha_1\,a\,b \\ \alpha_1 & 1/2 \end{bmatrix}$$

Then, R is equivalent to the block-diagonal matrix $R\_dec = V\,R\,U^{-1} \in B^{2\times 2}$ defined by:

```
> S:=HeuristicDecomposition(R,P,B):
> R_dec:=subs(alpha[1]^2=1/(4*a*b),map(collect,S[1],d[t]));
```

$$R\_dec := \begin{bmatrix} b\,d_x - \frac{d_t}{2\,\alpha_1} & 0 \\ 0 & -b\,d_x - \frac{d_t}{2\,\alpha_1} \end{bmatrix}$$

where the unimodular U and V are defined by:

```
> U:=simplify(subs(alpha[1]^2=1/(4*a*b),evalm(S[2])));
> V:=simplify(subs(alpha[1]^2=1/(4*a*b),evalm(S[3])));
```

$$U := \begin{bmatrix} -2\,\alpha_1 & 1 \\ 2\,\alpha_1 & 1 \end{bmatrix} \quad V := \begin{bmatrix} -2\,b\,\alpha_1 & 1 \\ 2\,b\,\alpha_1 & 1 \end{bmatrix}$$

If we substitute $\alpha_1 = 1/(2\sqrt{a\,b})$ into the previous block-diagonal matrix R_dec, we then obtain

```
> subs(alpha[1]=1/(2*sqrt(a*b)),evalm(R_dec));
```

$$\begin{bmatrix} b\,d_x - \sqrt{a\,b}\,d_t & 0 \\ 0 & b\,d_x + \sqrt{a\,b}\,d_t \end{bmatrix}$$

i.e., $R\_dec = \mathrm{diag}(b\, d_x - \sqrt{a\, b}\, d_t, b\, d_x + \sqrt{a\, b}\, d_t)$. The previous computations constitute an algebraic proof of the classical D'Alembert theorem (see, e.g., R. Courant, D. Hilbert, *Methods of Mathematical Physics*, Wiley Classics Library, Wiley, 1989).